

Vector calculus in Cartesian and spherical coordinates

This worksheet illustrates some features of [SageManifolds](#) (version 1.0, as included in SageMath 7.5) regarding vector calculus in the Euclidean 3-space by means of Cartesian and spherical coordinates.

Click [here](#) to download the worksheet file (ipynb format). To run it, you must start SageMath with the Jupyter notebook, via the command `sage -n jupyter`

NB: a version of SageMath at least equal to 7.5 is required to run this worksheet:

```
In [1]: version()
```

```
Out[1]: 'SageMath version 7.5, Release Date: 2017-01-11'
```

Euclidean 3-space and Cartesian coordinates

First we set up the notebook to display mathematical objects using LaTeX formatting:

```
In [2]: %display latex
```

We then introduce the Euclidean space as a 3-dimensional differentiable manifold:

```
In [3]: M = Manifold(3, 'M', start_index=1)
print(M)
```

```
3-dimensional differentiable manifold M
```

We then introduce the Cartesian coordinates (x, y, z) as the chart `cart` on M :

```
In [4]: cart.<x,y,z> = M.chart()
print(cart)
cart
```

```
Chart (M, (x, y, z))
```

```
Out[4]: (M, (x, y, z))
```

Spherical coordinates

We introduce spherical coordinates (r, θ, ϕ) as the chart `spher` on M :

```
In [5]: spher.<r,th,ph> = M.chart(r'r:(0,+oo) th:(0,pi):\theta ph:(0,2*pi):\phi
')
print(spher)
spher
```

```
Chart (M, (r, th, ph))
```

```
Out[5]: (M, (r, \theta, \phi))
```

Spherical coordinates do not form a regular coordinate system of the Euclidean space. So declaring that they span M means that, strictly speaking, the manifold M is not the whole Euclidean space, but the Euclidean space minus some half plane (the azimuthal origin). However, in this worksheet, this difference will not matter.

The change of coordinates $(r, \theta, \phi) \rightarrow (x, y, z)$ is introduced as a transition map from chart `spher` to chart `cart`:

```
In [6]: spher_to_cart = spher.transition_map(cart, [r*sin(th)*cos(ph), r*sin(th)
*spher_to_cart.display()
```

$$\text{Out[6]: } \begin{cases} x &= r \cos(\phi) \sin(\theta) \\ y &= r \sin(\phi) \sin(\theta) \\ z &= r \cos(\theta) \end{cases}$$

The inverse is also set:

```
In [7]: spher_to_cart.set_inverse(sqrt(x^2+y^2+z^2), atan2(sqrt(x^2+y^2),z), at
an2(y, x),
verbose=True)
```

Check of the inverse coordinate transformation:

```
r == r
th == arctan2(r*sin(th), r*cos(th))
ph == arctan2(r*sin(ph)*sin(th), r*cos(ph)*sin(th))
x == x
y == y
z == z
```

The check that the provided formulas do correspond to the inverse change of coordinates is passed, modulo some lack of simplification in some trigonometrical formulas involving the function `arctan2`.

```
In [8]: cart_to_spher = spher_to_cart.inverse()
cart_to_spher.display()
```

$$\text{Out[8]: } \begin{cases} r &= \sqrt{x^2 + y^2 + z^2} \\ \theta &= \arctan(\sqrt{x^2 + y^2}, z) \\ \phi &= \arctan(y, x) \end{cases}$$

The natural vector frame of spherical coordinates is

```
In [9]: spher.frame()
```

$$\text{Out[9]: } \left(M, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right)$$

We shall expand vector and tensor fields on the orthonormal frame (e_1, e_2, e_3) associated with spherical coordinates, which is related to the natural frame $(\partial/\partial r, \partial/\partial \theta, \partial/\partial \phi)$ displayed above by means of the following field of automorphisms:

```
In [10]: to_orthonormal = M.automorphism_field()
to_orthonormal[spher.frame(),1,1,spher] = 1
to_orthonormal[spher.frame(),2,2,spher] = 1/r
to_orthonormal[spher.frame(),3,3,spher] = 1/(r*sin(th))
to_orthonormal.display(spher.frame(), spher)
```

$$\text{Out[10]: } \frac{\partial}{\partial r} \otimes dr + \frac{1}{r} \frac{\partial}{\partial \theta} \otimes d\theta + \frac{1}{r \sin(\theta)} \frac{\partial}{\partial \phi} \otimes d\phi$$

In other words, the change-of-basis matrix is

```
In [11]: to_orthonormal[spher.frame(),:,spher]
```

$$\text{Out[11]: } \begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{r} & 0 \\ 0 & 0 & \frac{1}{r \sin(\theta)} \end{pmatrix}$$

We construct the orthonormal frame from the natural frame of spherical coordinates by this change of basis:

```
In [12]: es = spher.frame().new_frame(to_orthonormal, 'e')
print(es)
es
```

Vector frame (M, (e_1,e_2,e_3))

$$\text{Out[12]: } (M, (e_1, e_2, e_3))$$

```
In [13]: es[1].display(spher.frame(), spher)
```

$$\text{Out[13]: } e_1 = \frac{\partial}{\partial r}$$

```
In [14]: es[2].display(spher.frame(), spher)
```

$$\text{Out[14]: } e_2 = \frac{1}{r} \frac{\partial}{\partial \theta}$$

```
In [15]: es[3].display(spher.frame(), spher)
```

$$\text{Out[15]: } e_3 = \frac{1}{r \sin(\theta)} \frac{\partial}{\partial \phi}$$

If we do not specify the frame and coordinates for the display, we get it in terms of the default ones (i.e. Cartesian frame and Cartesian coordinates):

```
In [16]: es[1].display()
```

$$\text{Out[16]: } e_1 = \left(\frac{x}{\sqrt{x^2 + y^2 + z^2}} \right) \frac{\partial}{\partial x} + \left(\frac{y}{\sqrt{x^2 + y^2 + z^2}} \right) \frac{\partial}{\partial y} + \left(\frac{z}{\sqrt{x^2 + y^2 + z^2}} \right) \frac{\partial}{\partial z}$$

In [17]: `es[2].display()`

Out[17]:

$$e_2 = \left(\frac{xz}{\sqrt{x^2 + y^2 + z^2} \sqrt{x^2 + y^2}} \right) \frac{\partial}{\partial x} + \left(\frac{yz}{\sqrt{x^2 + y^2 + z^2} \sqrt{x^2 + y^2}} \right) \frac{\partial}{\partial y} + \left(-\frac{\sqrt{x^2 + y^2}}{\sqrt{x^2 + y^2 + z^2}} \right) \frac{\partial}{\partial z}$$

In [18]: `es[3].display()`

Out[18]:

$$e_3 = \left(-\frac{y}{\sqrt{x^2 + y^2}} \right) \frac{\partial}{\partial x} + \left(\frac{x}{\sqrt{x^2 + y^2}} \right) \frac{\partial}{\partial y}$$

By construction, the change of frame $(\partial/\partial r, \partial/\partial \theta, \partial/\partial \phi) \rightarrow (e_1, e_2, e_3)$ is known. We form the change of frame $(\partial/\partial x, \partial/\partial y, \partial/\partial z) \rightarrow (e_1, e_2, e_3)$ by composition of $(\partial/\partial x, \partial/\partial y, \partial/\partial z) \rightarrow (\partial/\partial r, \partial/\partial \theta, \partial/\partial \phi)$ with $(\partial/\partial r, \partial/\partial \theta, \partial/\partial \phi) \rightarrow (e_1, e_2, e_3)$:

In [19]:

```
M.set_change_of_frame(cart.frame(), es,
                      M.change_of_frame(spher.frame(), es) * M.change_of_frame(cart.frame(),
                      spher.frame()),
                      compute_inverse=False)
M.change_of_frame(cart.frame(), es)[: , spher]
```

Out[19]:

$$\begin{pmatrix} \cos(\phi) \sin(\theta) & \cos(\phi) \cos(\theta) & -\sin(\phi) \\ \sin(\phi) \sin(\theta) & \cos(\theta) \sin(\phi) & \cos(\phi) \\ \cos(\theta) & -\sin(\theta) & 0 \end{pmatrix}$$

Similarly, we form the inverse change of frame as:

In [20]:

```
M.set_change_of_frame(es, cart.frame(),
                      M.change_of_frame(spher.frame(), cart.frame()) *
                      M.change_of_frame(es, spher.frame()),
                      compute_inverse=False)
M.change_of_frame(es, cart.frame())[: , spher]
```

Out[20]:

$$\begin{pmatrix} \cos(\phi) \sin(\theta) & \sin(\phi) \sin(\theta) & \cos(\theta) \\ \cos(\phi) \cos(\theta) & \cos(\theta) \sin(\phi) & -\sin(\theta) \\ -\sin(\phi) & \cos(\phi) & 0 \end{pmatrix}$$

At this stage, the manifold (user) atlas is

In [21]: `M.atlas()`

Out[21]: $[(M, (x, y, z)), (M, (r, \theta, \phi))]$

The default chart is the first one introduced on the manifold (it can be changed by means of the function `M.set_default_chart`):

In [22]: `M.default_chart()`

Out[22]: $(M, (x, y, z))$

The following vector frames have been introduced on the manifold:

In [23]: `M.frames()`

Out[23]: $\left[\left(M, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right), \left(M, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right), (M, (e_1, e_2, e_3)) \right]$

The default frame is the first one introduced on the manifold (it can be changed by means of the function `M.set_default_frame`):

In [24]: `M.default_frame()`

Out[24]: $\left(M, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right)$

The following changes of frame have been defined:

In [25]: `M.changes_of_frame()`

Out[25]:
$$\left\{ \left(\left(M, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right), (M, (e_1, e_2, e_3)) \right) \right.$$

: Field of tangent-space automorphisms on the 3-dimensional differentiable manifold \mathbb{R}^3

$$\left((M, (e_1, e_2, e_3)), \left(M, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right) \right)$$

: Field of tangent-space automorphisms on the 3-dimensional differentiable manifold \mathbb{R}^3

$$\left(\left(M, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right), \left(M, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right) \right)$$

: Field of tangent-space automorphisms on the 3-dimensional differentiable manifold \mathbb{R}^3

$$\left(\left(M, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right), \left(M, \left(\frac{\partial}{\partial x}, \frac{\partial}{\partial y}, \frac{\partial}{\partial z} \right) \right) \right)$$

: Field of tangent-space automorphisms on the 3-dimensional differentiable manifold \mathbb{R}^3

$$\left((M, (e_1, e_2, e_3)), \left(M, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right) \right)$$

: Field of tangent-space automorphisms on the 3-dimensional differentiable manifold \mathbb{R}^3

$$\left(\left(M, \left(\frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi} \right) \right), (M, (e_1, e_2, e_3)) \right)$$

: Field of tangent-space automorphisms on the 3-dimensional differentiable manifold \mathbb{R}^3

$$\left. \right\}$$

Vector field defined in terms of its Cartesian components

We define the vector field U in terms of its components with respect to the default frame, i.e. the Cartesian frame:

In [26]: `U = M.vector_field(name='U')
U[:] = [function('U_x')(x,y,z), function('U_y')(x,y,z), function('U_z')(x,y,z)]
U.display()`

Out[26]: $U = U_x(x, y, z) \frac{\partial}{\partial x} + U_y(x, y, z) \frac{\partial}{\partial y} + U_z(x, y, z) \frac{\partial}{\partial z}$

We can ask for its components in terms of the spherical orthonormal frame:

In [27]: `U.display(es)`

Out[27]:

$$U = \left(\frac{xU_x(x, y, z) + yU_y(x, y, z) + zU_z(x, y, z)}{\sqrt{x^2 + y^2 + z^2}} \right) e_1 \\ + \left(-\frac{x^2U_z(x, y, z) + y^2U_z(x, y, z) - (xU_x(x, y, z) + yU_y(x, y, z))z}{\sqrt{x^2 + y^2 + z^2}\sqrt{x^2 + y^2}} \right) e_2 \\ + \left(-\frac{yU_x(x, y, z) - xU_y(x, y, z)}{\sqrt{x^2 + y^2}} \right) e_3$$

In [28]: `U.display_comp(es)`

Out[28]:

$$U^1 = \frac{xU_x(x, y, z) + yU_y(x, y, z) + zU_z(x, y, z)}{\sqrt{x^2 + y^2 + z^2}} \\ U^2 = -\frac{x^2U_z(x, y, z) + y^2U_z(x, y, z) - (xU_x(x, y, z) + yU_y(x, y, z))z}{\sqrt{x^2 + y^2 + z^2}\sqrt{x^2 + y^2}} \\ U^3 = -\frac{yU_x(x, y, z) - xU_y(x, y, z)}{\sqrt{x^2 + y^2}}$$

The above components are displayed in terms of the default chart (Cartesian coordinates). If we want them in terms of spherical coordinates, we have to specify it, by setting the second argument of the function `display` to `spher`:

In [29]: `U.display(es, spher)`

Out[29]:

$$U \\ = (U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \sin(\theta) \\ + U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\phi) \sin(\theta) \\ + U_z(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\theta)) e_1 \\ + (U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \cos(\theta) \\ + U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\theta) \sin(\phi) \\ - U_z(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\theta)) e_2 \\ + (U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \\ - U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\phi)) e_3$$

In [30]: `U.display_comp(es, spher)`

Out[30]:

$$U^1 = U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \sin(\theta) \\ + U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\phi) \sin(\theta) \\ + U_z(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\theta) \\ U^2 = U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \cos(\theta) \\ + U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\theta) \sin(\phi) \\ - U_z(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\theta) \\ U^3 = U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \\ - U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\phi)$$

We may also ask for the components of U w.r.t. the natural frame of spherical coordinates:

In [31]: `U.display(spher.frame())`

Out[31]:

$$U = \left(\frac{xU_x(x, y, z) + yU_y(x, y, z) + zU_z(x, y, z)}{\sqrt{x^2 + y^2 + z^2}} \right) \frac{\partial}{\partial r} + \left(-\frac{(x^2U_z(x, y, z) + y^2U_z(x, y, z) - (xU_x(x, y, z) + yU_y(x, y, z))z)\sqrt{x^2 + y^2}}{x^4 + 2x^2y^2 + y^4 + (x^2 + y^2)z^2} \right) \frac{\partial}{\partial \theta} + \left(-\frac{yU_x(x, y, z) - xU_y(x, y, z)}{x^2 + y^2} \right) \frac{\partial}{\partial \phi}$$

In [32]: `U.display(spher.frame(), spher)`

Out[32]:

$$U = (U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \sin(\theta) + U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\phi) \sin(\theta) + U_z(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\theta)) \frac{\partial}{\partial r} + \left(\frac{U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) \cos(\theta) + U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\theta) \sin(\phi) - U_z(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\theta)}{r} \right) \frac{\partial}{\partial \theta} + \left(\frac{U_y(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \cos(\phi) - U_x(r \cos(\phi) \sin(\theta), r \sin(\phi) \sin(\theta), r \cos(\theta)) \sin(\phi)}{r \sin(\theta)} \right) \frac{\partial}{\partial \phi}$$

Vector field defined in terms of its spherical components

Let us consider a vector field V defined by its components with respect to the orthonormal spherical frame $es = (e_1, e_2, e_3)$:

In [33]: `V = M.vector_field(name='V')
V[es, :, spher] = [function('V_1')(r, th, ph), function('V_2')(r, th, ph), function('V_3')(r, th, ph)]
V.display(es, spher)`

Out[33]: $V = V_1(r, \theta, \phi) e_1 + V_2(r, \theta, \phi) e_2 + V_3(r, \theta, \phi) e_3$

We may ask for the components of this vector field with respect to the Cartesian frame (`cart.frame()`), each component being expressed in terms of spherical coordinates (second argument `spher`):

```
In [34]: V.display(cart.frame(), spher)
```

$$\begin{aligned} \text{Out [34]: } V = & (V_2(r, \theta, \phi) \cos(\phi) \cos(\theta) + V_1(r, \theta, \phi) \cos(\phi) \sin(\theta) - V_3(r, \theta, \phi) \sin(\phi)) \frac{\partial}{\partial x} \\ & + (V_2(r, \theta, \phi) \cos(\theta) \sin(\phi) + V_1(r, \theta, \phi) \sin(\phi) \sin(\theta) + V_3(r, \theta, \phi) \cos(\phi)) \frac{\partial}{\partial y} \\ & + (V_1(r, \theta, \phi) \cos(\theta) - V_2(r, \theta, \phi) \sin(\theta)) \frac{\partial}{\partial z} \end{aligned}$$

```
In [35]: V.display_comp(cart.frame(), spher)
```

$$\begin{aligned} \text{Out [35]: } V^x &= V_2(r, \theta, \phi) \cos(\phi) \cos(\theta) + V_1(r, \theta, \phi) \cos(\phi) \sin(\theta) - V_3(r, \theta, \phi) \sin(\phi) \\ V^y &= V_2(r, \theta, \phi) \cos(\theta) \sin(\phi) + V_1(r, \theta, \phi) \sin(\phi) \sin(\theta) + V_3(r, \theta, \phi) \cos(\phi) \\ V^z &= V_1(r, \theta, \phi) \cos(\theta) - V_2(r, \theta, \phi) \sin(\theta) \end{aligned}$$

Euclidean metric

The standard Euclidean metric is introduced as a Riemannian metric on M , whose components with respect to the Cartesian frame are $\text{diag}(1, 1, 1)$:

```
In [36]: g = M.riemannian_metric('g')
g[1,1], g[2,2], g[3,3] = 1, 1, 1
g.display()
```

$$\text{Out [36]: } g = dx \otimes dx + dy \otimes dy + dz \otimes dz$$

The components of g with respect to the spherical coordinates are then:

```
In [37]: g.display(spher.frame(), spher)
```

$$\text{Out [37]: } g = dr \otimes dr + r^2 d\theta \otimes d\theta + r^2 \sin(\theta)^2 d\phi \otimes d\phi$$

Since $e_s = (e_1, e_2, e_3)$ is an orthonormal frame, the components of g with respect to it are $\text{diag}(1, 1, 1)$:

```
In [38]: g.display(es, spher)
```

$$\text{Out [38]: } g = e^1 \otimes e^1 + e^2 \otimes e^2 + e^3 \otimes e^3$$

The covariant derivative operator ∇ is introduced as the (Levi-Civita) connection associated with g :

```
In [39]: nabla = g.connection()
print(nabla)
nabla
```

Levi-Civita connection `nabla_g` associated with the Riemannian metric `g` on the 3-dimensional differentiable manifold `M`

$$\text{Out [39]: } \nabla_g$$

The connection coefficient with respect to the natural frame of spherical coordinates (Christoffel symbols) are:


```
In [40]: nabl.display(spher.frame(), spher)
```

```
Out[40]:
```

$$\begin{aligned}\Gamma^r_{\theta\theta} &= -r \\ \Gamma^r_{\phi\phi} &= -r \sin(\theta)^2 \\ \Gamma^\theta_{r\theta} &= \frac{1}{r} \\ \Gamma^\theta_{\theta r} &= \frac{1}{r} \\ \Gamma^\theta_{\phi\phi} &= -\cos(\theta) \sin(\theta) \\ \Gamma^\phi_{r\phi} &= \frac{1}{r} \\ \Gamma^\phi_{\theta\phi} &= \frac{\cos(\theta)}{\sin(\theta)} \\ \Gamma^\phi_{\phi r} &= \frac{1}{r} \\ \Gamma^\phi_{\phi\theta} &= \frac{\cos(\theta)}{\sin(\theta)}\end{aligned}$$

while those with respect to the orthonormal spherical frame are:

```
In [41]: nabl.display(es, spher)
```

```
Out[41]:
```

$$\begin{aligned}\Gamma^1_{22} &= -\frac{1}{r} \\ \Gamma^1_{33} &= -\frac{1}{r} \\ \Gamma^2_{12} &= \frac{1}{r} \\ \Gamma^2_{33} &= -\frac{\cos(\theta)}{r \sin(\theta)} \\ \Gamma^3_{13} &= \frac{1}{r} \\ \Gamma^3_{23} &= \frac{\cos(\theta)}{r \sin(\theta)}\end{aligned}$$

The covariant derivative of U is

```
In [42]: nabU = nabl(U)
print(nabU)
```

Tensor field nabl_g(U) of type (1,1) on the 3-dimensional differentiable manifold M

```
In [43]: nabU.display()
```

```
Out[43]:
```

$$\begin{aligned}\nabla_g U &= \frac{\partial U_x}{\partial x} \frac{\partial}{\partial x} \otimes dx + \frac{\partial U_x}{\partial y} \frac{\partial}{\partial x} \otimes dy + \frac{\partial U_x}{\partial z} \frac{\partial}{\partial x} \otimes dz + \frac{\partial U_y}{\partial x} \frac{\partial}{\partial y} \otimes dx \\ &+ \frac{\partial U_y}{\partial y} \frac{\partial}{\partial y} \otimes dy + \frac{\partial U_y}{\partial z} \frac{\partial}{\partial y} \otimes dz + \frac{\partial U_z}{\partial x} \frac{\partial}{\partial z} \otimes dx + \frac{\partial U_z}{\partial y} \frac{\partial}{\partial z} \otimes dy + \frac{\partial U_z}{\partial z} \frac{\partial}{\partial z} \\ &\otimes dz\end{aligned}$$

while the covariant derivative of V is

```
In [44]: nabV = nabl(V)
print(nabV)
```

Tensor field nabl_g(V) of type (1,1) on the 3-dimensional differentiable manifold M

In [45]: `nabV.display(es, spher)`

Out[45]:

$$\begin{aligned} \nabla_g V = & \frac{\partial V_1}{\partial r} e_1 \otimes e^1 + \left(-\frac{V_2(r, \theta, \phi) - \frac{\partial V_1}{\partial \theta}}{r} \right) e_1 \otimes e^2 \\ & + \left(-\frac{V_3(r, \theta, \phi) \sin(\theta) - \frac{\partial V_1}{\partial \phi}}{r \sin(\theta)} \right) e_1 \otimes e^3 + \frac{\partial V_2}{\partial r} e_2 \otimes e^1 \\ & + \left(\frac{V_1(r, \theta, \phi) + \frac{\partial V_2}{\partial \theta}}{r} \right) e_2 \otimes e^2 + \left(-\frac{V_3(r, \theta, \phi) \cos(\theta) - \frac{\partial V_2}{\partial \phi}}{r \sin(\theta)} \right) e_2 \otimes e^3 \\ & + \frac{\partial V_3}{\partial r} e_3 \otimes e^1 + \frac{\frac{\partial V_3}{\partial \theta}}{r} e_3 \otimes e^2 \\ & + \left(\frac{V_2(r, \theta, \phi) \cos(\theta) + V_1(r, \theta, \phi) \sin(\theta) + \frac{\partial V_3}{\partial \phi}}{r \sin(\theta)} \right) e_3 \otimes e^3 \end{aligned}$$

In []: